# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=cerbere
pkgver=2.5.1
pkgrel=2
pkgdesc='The Pantheon Watchdog'
arch=(loong64 x86_64)
url=https://github.com/elementary/cerbere
license=(GPL2)
groups=(pantheon)
depends=(
  dconf
  glib2
  libgee
)
makedepends=(
  git
  meson
  vala
)
source=(git+https://github.com/elementary/cerbere.git#tag=fcfe622aea6f50c23b1aed4a5bf20e6c515c31c1)
sha256sums=(SKIP)

pkgver() {
  cd cerbere

  git describe --tags
}

build() {
  arch-meson cerbere build
  ninja -C build
}

package() {
  DESTDIR="${pkgdir}" meson install -C build
}

# vim: ts=2 sw=2 et:
